
INTRODUCTION
This application note discusses the use of the
PIC16CXXX Timer1 module as an asynchronous clock.
The Timer1 module has it own oscillator circuitry, which
allows the timer to keep real-time, even when the
device is in SLEEP mode. When the device is in sleep,
the oscillator will continue to increment TMR1. An over-
flow of the TMR1 register causes a TMR1 Overflow
Interrupt (if enabled) and will wake the processor from
sleep. The interrupt service routine can then perform
the desired task. 

OVERVIEW
Timer1 is a 16-bit counter with a 2-bit prescaler. Timer1
can be incremented from an internal clock, an external
clock, or an external oscillator. Timer1 can be
configured to synchronize or not synchronize the
external clock sources. Asynchronous operation allows
Timer1 to increment when the device is in sleep.
Figure 1 is a block diagram of Timer1.

To set up Timer1 for asynchronous operation the
Timer1 control register, T1CON, must have the follow-
ing bits configured:

• TMR1CS set (external clock source)
• T1CKS1:T1CKS0 configured for the desired 

prescaler
• T1SYNC set (asynchronous operation)
• TMR1ON set (enables Timer1)
• T1OSCEN set, if using an external oscillator

Author: Mark Palmer
Microchip Technology Inc.
  1997 Microchip Technology Inc.
In asynchronous operation, if the clock source is an
external clock, it is input on the T1CKI pin. If the clock
source is a crystal oscillator, the crystal is connected
across the T1OSO and T1OSI pins. 

When using Timer1 in Asynchronous mode, the use of
an external clock minimizes the operating and sleep
currents. This is because the timer’s internal oscillator
circuitry is disabled. Though the external clock may give
the lower device currents, the use of a crystal oscillator
may lead to lower system current consumption and sys-
tem cost.

System current consumption can also be reduced by
having the TMR1 Overflow Interrupt wake the
processor from SLEEP at the desired interval, With a
32.768 kHz crystal, Timer1’s overflow rate ranges from
2 to 16 seconds, depending on the prescaler chosen.
Table 1 shows Timer1 overflow times for various crystal
frequencies and prescaler values.

TABLE 1: TIMER1 OVERFLOW TIMES

Frequency (kHz)

Prescale 32.768 100 200

1 2 0.655 0.327

2 4 1.31 0.655

4 8 2.62 1.31

8 16 5.24 2.62

Overflow times in seconds.
FIGURE 1: TIMER1 BLOCK DIAGRAM

As can be seen the 32 kHz crystal, gives very nice
overflow rates. These crystals, referred to as watch
crystals, also can be relatively inexpensive. In many
applications the 2 second overflow time, of a 32 kHz
crystal, is too long. An easy way to reduce the overflow
time is to load the TMR1H register with a value, during
the interrupt service routine. Table 2 shows the over-
flow times, depending on the value loaded into the
TMR1H register and a prescale of 1.

TABLE 2: TMR1H LOAD VALUES / TIMER1 
OVERFLOW TIMES

TMR1H Load Value
Overflow Time
(@ 32.768 kHz)

80h 1 Second

C0h 0.5 Second

E0h 0.25 Second

F0h 0.125 Second

Note: The loading of either TMR1H or TMR1L
causes the prescaler to be cleared. When
Timer1 is in operation, extreme care
should be taken in modifying either the
TMR1H or TMR1L registers, since this
automatically configures the prescaler
to 1.
DS00580C-page  2
The code segment shown in Example 1 configures the
Timer1 module for asynchronous operation, enables
the Timer1 interrupt, and the interrupt service routine
loads the TMR1H register with a value.

CONCLUSION

Timer1 gives designers a powerful time-base function.
The asynchronous operation and internal oscillator cir-
cuitry gives designers the ability to easily keep real-
time, while minimizing power consumption and external
logic. 
  1997 Microchip Technology Inc.

EXAMPLE 1: TIMER1 CODE SEGMENT FOR ASYNCHRONOUS OPERATION

      org   0x000
Reset_V     GOTO    START
;
      org   0x004
PER_INT_V
            BCF     STATUS, RP0    ; Bank0
            BTFSC   PIR1, TMR1IF   ; Timer1 overflowed?
            GOTO    T1_OVRFL       ; YES, Service the Timer1 Overflow Interrupt
;
; Should NEVER get here
;
ERROR1                             ; NO, Unknown Interrupt Source
            BSF     PORTD, 1       ; Toggle a port pin to indicate error
            BCF     PORTD, 1
            GOTO    ERROR1
;
T1_OVRFL
            BCF     PIR1, TMR1IF   ; Clear Timer1 Interrupt Flag
            MOVLW   0x80           ; Since doing key inputs, clear TMR1
            MOVWF   TMR1H          ;   for 1 sec overflow.
            :
            :                      ; Do Interrupt stuff here
            :
            RETFIE                 ; Return / Enable Global Interrupts
;
;
START                              ; POWER_ON Reset (Beginning of program)
            CLRF    STATUS         ; Do initialization (Bank0)
            BCF     T1CON, TMR1ON  ; Timer1 is NOT incrementing
            :
            :                      ; Do Initialization stuff here
            :
            MOVLW   0x80           ; TIM1H:TMR1L = 0x8000 gives 1 second
            MOVWF   TMR1H          ;   overflow, at 32 KHz.
            CLRF    TMR1L          ;
;
            CLRF    INTCON
            CLRF    PIR1
            BSF     STATUS, RP0    ; Bank1
            CLRF    PIE1           ; Disable all peripheral interrupts
;
    if ( C74_REV_A )               ; See PIC16C74 Errata
            BSF     TRISC, T1OSO   ; RC0 needs to be input for the oscillator to function
    endif
            BSF     PIE1, TMR1IE   ; Enable TMR1 Interrupt
;
; Initialize the Special Function Registers (SFR) interrupts
;
            BCF     STATUS, RP0    ; Bank0
            CLRF    PIR1           ;
            BSF     INTCON, PEIE   ; Enable Peripheral Interrupts
            BSF     INTCON, GIE    ; Enable all Interrupts
;
            MOVLW   0x0E
            MOVWF   T1CON          ; Enable T1 Oscillator, Ext Clock, Async, prescaler = 1
            BSF     T1CON, TMR1ON  ; Turn Timer1 ON
;
zzz         SLEEP
            GOTO    zzz            ; Sleep, wait for TMR1 interrupt

